La elipse de los baricentros

Francisco Javier García Capitán

1. Introducción

Es conocido que las medianas de un triángulo dividen a éste en seis triángulos de la misma área, y que los baricentros de estos seis triángulos están en una elipse centrada en el baricentro del triángulo de partida[1].



A esta elipse la llamaremos *la elipse de los baricentros* del triángulo.

Aunque una cónica queda determinada por cinco puntos, cabe preguntarse para qué triángulos la elipse de los baricentros contiene también a otros puntos notables del triángulo, como el circuncentro, el ortocentro o el incentro.

En nuestra investigación haremos uso del programa de cálculo simbólico *Mathematica*.

[1]En efecto, dado un triángulo T, basta considerar una transformación afín ϕ que transforme T en un triángulo equilátero E. Las transformaciones afines convierten medianas en medianas, baricentros en baricentros, elipses en elipses y centros de elipses en centros de elipses. En el triángulo equilátero E, los seis triángulos formados por las medianas son congruentes, y sus baricentros están a la misma distancia del centro de E, estando por tanto en una circunferencia centrada en el baricentro de E. Sus imágenes por ϕ^{-1} estarán en una elipse con centro el baricentro de T.

En todo lo que sigue consideraremos fijo un segmento AB, con coordenadas $A = (-1, 0)$, $B = (1, 0)$ y variable un tercer punto $C = (u, v)$. Buscaremos condiciones sobre u y v para que el circuncentro, ortocentro o incentro del triángulo ABC pertenezcan a la elipse de los baricentros de ABC.

Para ello, hallaremos la ecuación de la elipse de los baricentros para este triángulo, así como las coordenadas de los puntos notables, y después sustituiremos las coordenadas en la ecuación, resultando la ecuación de los lugares geométricos buscados.

2. La elipse de los baricentros

Comencemos por hallar la ecuación de la elipse de los baricentros de nuestro triángulo. Necesitamos algunas fórmulas preliminares:

- Punto medio de un segmento:

$$\texttt{PuntoMedio[\{x1\_, y1\_\}, \{x2\_, y2\_\}] :=} \left\{\frac{x1 + x2}{2}, \frac{y1 + y2}{2}\right\};$$

- Baricentro de un triángulo:

$$\texttt{Baricentro[\{x1\_, y1\_\}, \{x2\_, y2\_\}, \{x3\_, y3\_\}] :=} \left\{\frac{x1 + x2 + x3}{3}, \frac{y1 + y2 + y3}{3}\right\};$$

- Cónica que pasa por cinco puntos:

$$\texttt{ConicaCincoPuntos[\{x1\_, y1\_\}, \{x2\_, y2\_\}, \{x3\_, y3\_\}, \{x4\_, y4\_\}, \{x5\_, y5\_\}] :=}$$
$$\texttt{Det}\left[\begin{pmatrix} x^2 & x\,y & y^2 & x & y & 1 \\ x1^2 & x1\,y1 & y1^2 & x1 & y1 & 1 \\ x2^2 & x2\,y2 & y2^2 & x2 & y2 & 1 \\ x3^2 & x3\,y3 & y3^2 & x3 & y3 & 1 \\ x4^2 & x4\,y4 & y4^2 & x4 & y4 & 1 \\ x5^2 & x5\,y5 & y5^2 & x5 & y5 & 1 \end{pmatrix}\right];$$

Ya podemos hallar la ecuación de nuestra elipse:

```
ElipseBaricentros[p1_, p2_, p3_] :=
  Module[{m1, m2, m3, g, g1, g2, g3, g4, g5, g6},
   {m1 := PuntoMedio[p2, p3];
    m2 := PuntoMedio[p3, p1];
    m3 := PuntoMedio[p1, p2];
    g  := Baricentro[p1, p2, p3];
    g1 := Baricentro[g, p1, m3];
    g2 := Baricentro[g, p1, m2];
    g3 := Baricentro[g, p2, m1];
    g4 := Baricentro[g, p2, m3];
    g5 := Baricentro[g, p3, m1];
    g6 := Baricentro[g, p3, m2];
    ConicaCincoPuntos[g2, g3, g4, g5, g6]
    }];
```

Aplicando la función `ElipseBaricentros` a nuestro triángulo obtendremos la ecuación de su elipse de los baricentros:

$$27v^2x^2 + (27u^2 + 81)y^2 - 54vy - 54uvxy + 2v^2 = 0.$$

3. El circuncentro

Queremos hallar ahora los puntos C tales que el circuncentro O del triángulo ABC está en la elipse de los baricentros. Usamos *Mathematica* de nuevo para hallar las coordenadas del circuncentro.

```
Mediatriz[{x1_, y1_}, {x2_, y2_}] :=
  (x1 - x2) (x - (x1 + x2)/2) + (y1 - y2) (y - (y1 + y2)/2) == 0;

Circuncentro[p1_, p2_, p3_] := Solve[
    {Mediatriz[p1, p2], Mediatriz[p1, p3]}, {x, y}][[1]];

Circuncentro[{-1, 0}, {1, 0}, {u, v}]
```

El resultado es el punto $\left(0, \frac{u^2+v^2-1}{2v}\right)$. Le decimos a *Mathematica* que sustituya estas coordenadas en la ecuación de la elipse:

```
Factor[ElipseBaricentros[{-1, 0}, {1, 0}, {u, v}] /.
  Circuncentro[{-1, 0}, {1, 0}, {u, v}]]
```

Si en la fórmula obtenida cambiamos las variables u y v por x e y, respectivamente, resulta que la ecuación del lugar geométrico buscado es

$$27x^2y^4 - 19y^4 + 54x^4y^2 - 54y^2 + 27x^6 + 27x^4 - 135x^2 + 81 = 0,$$

o en forma explícita:

$$y = \pm\sqrt{\frac{27 - 27x^4 \pm 6(1 - x^2)\sqrt{63 - 6x^2}}{27x^2 - 19}}.$$

La figura adjunta muestra, el triángulo ABC para una de las soluciones, en las que vemos el circuncentro O en la elipse de los baricentros.



El lugar geométrico del punto C es una curva, dibujada en la figura con trazo grueso, compuesta por una "casi circunferencia" y dos trozos de curva que son asintóticos a sendas rectas paralelas. Es curioso lo próxima que está la "casi circunferencia" a ser la circunferencia con diámetro AB, que aparece punteada en la figura, así como la pequeña curvatura de los otros dos trozos.

Las ecuaciones de las asíntotas son

$$x = \pm\sqrt{\frac{19}{27}} = \pm 0.83887.$$

El valor máximo de y en la "casi circunferencia" es

$$\sqrt{\frac{18\sqrt{7} - 27}{19}} = 1{,}04185,$$

que indica lo cerca que está la "casi circunferencia" de ser una verdadera circunferencia.

4. El ortocentro

Procedemos de forma similar para obtener la ecuación del lugar geométrico correspondiente al ortocentro:

```
Altura[{x1_, y1_}, {x2_, y2_}, {x3_, y3_}] :=
  (x2 - x3) (x - x1) + (y2 - y3) (y - y1) == 0;

Ortocentro[p1_, p2_, p3_] := Solve[
    {Altura[p1, p2, p3], Altura[p2, p3, p1]}, {x, y}][[1]];

Factor[ElipseBaricentros[{-1, 0}, {1, 0}, {u, v}] /.
  Ortocentro[{-1, 0}, {1, 0}, {u, v}]]
```

La ecuación del lugar geométrico es ahora

$$27x^2y^4 + 2y^4 + 54x^4y^2 - 54y^2 + 27x^6 + 27x^4 - 135x^2 + 81 = 0.$$

La ecuación obtenida es muy parecida a la correspondiente al circuncentro.



En la figura adjunta, donde hemos trazado la curva, se muestra que la gráfica es sensiblemente diferente. Vemos un triángulo ABC y su ortocentro H que pertenece a la elipse de los baricentros. La circunferencia con diámetro AB se muestra con trazo punteado.

La curva está formada por dos trozos "casi triangulares". Uno de los lados de los triángulos es "casi paralelo" a la circunferencia con diámetro AB.

En forma explícita, el lugar geométrico tiene la ecuación

$$y = \pm\sqrt{\frac{27 - 27x^4 \pm 3\sqrt{3(21 - 71x^2 + 79x^4 - 29x^6)}}{27x^2 + 2}},$$

fórmula que es válida para los valores

$$-\sqrt{\frac{21}{29}} \leqslant x \leqslant \sqrt{\frac{21}{29}}.$$

5. El incentro

Podemos esperar que el lugar geométrico correspondiente al incentro tenga una ecuación más complicada que las anteriores, ya que la fórmula del incentro incluye raíces cuadradas, que no estaban presentes en el circuncentro y en el ortocentro.

Vamos a usar las coordenadas baricéntricas para hallar las coordenadas cartesianas del incentro. El incentro tiene coordenadas baricéntricas $(a : b : c)$ siendo a, b y c las longitudes de los lados del triángulo ABC.

```
DeBaricentricasACartesianas[{u_, v_, w_},
  {xA_, yA_}, {xB_, yB_}, {xC_, yC_}] :=
 Factor[{x -> (u xA + v xB + w xC)/(u + v + w), y -> (u yA + v yB + w yC)/(u + v + w)} /.
   {a → Sqrt[(xB - xC)^2 + (yB - yC)^2],
    b → Sqrt[(xC - xA)^2 + (yC - yA)^2],
    c → Sqrt[(xA - xB)^2 + (yA - yB)^2]}]

Incentro[ptA_, ptB_, ptC_] := DeBaricentricasACartesianas[
   {a, b, c}, ptA, ptB, ptC];

Factor[ElipseBaricentros[{-1, 0}, {1, 0}, {u, v}] /.
  Incentro[{-1, 0}, {1, 0}, {u, v}]]
```

Como habíamos previsto, la ecuación del lugar geométrico es más complicada que las anteriores:

$$
\begin{gathered}
25\sqrt{y^2+(x+1)^2}\sqrt{y^2+(x-1)^2}-50\sqrt{y^2+(x+1)^2} \\
-50\sqrt{y^2+(x-1)^2}-29(x^2+y^2+3)=0.
\end{gathered}
$$

Para describir la curva obtenida nos vemos obligados a usar de nuevo el “casi”. Está formada por tres “casi circunferencias”. Dos de ellas están “centradas”en los puntos A y B y aparecen con trazo grueso la figura siguiente. Observamos que el incentro está en la elipse de los baricentros.



La tercera “casi circunferencia” está “centrada” en el punto medio de AB y tiene un “radio” mucho mayor. En la siguiente figura se observa su gran tamaño, pues las dos primeras “casi circunferencias” parecen reducirse a puntos:



La circunferencia que tiene su centro en el punto medio de AB y pasa por uno de los puntos del lugar geométrico, dibujada con trazo punteado, nos da una idea de lo próxima que está la curva a ser una circunferencia.

6. Conclusión

La investigación nos ha llevado al descubrimiento de unas interesantes curvas en las que podemos observar parecidos asombrosos con circunferencias rectas o triángulos.

El método usado para hallar el lugar geométrico correspondiente al incentro puede hacerse extensivo a cualquier punto relacionado del triángulo del que conozcamos sus ecuaciones baricéntricas, y en la enciclopedia de Clark Kimberling los tenemos a miles.

Como último ejemplo, ofrecemos el caso del punto simediano, cuyas coordenadas baricéntricas son $(a^2 : b^2 : c^2)$, y de nuevo nos sorprende el resultado:

¿quién diría que el lugar geométrico obtenido no está formado por dos circunferencias perfectas?



En este caso la ecuación del lugar geométrico de C es

$$x^4 + y^4 + 2x^2y^2 + 6x^2 - 48y^2 + 9 = 0,$$

que es la unión de las circunferencias $x^2 + y^2 \pm 3\sqrt{6}y + 3 = 0$.

Francisco Javier García Capitán, 2004
pacoga@ctv.es